

Paper BI-01-2013

Using the SAS® Projman Application for Scheduling Projects

Jon Patton, Miami University, Oxford, Ohio

ABSTRACT

SAS/OR software has four major procedures that can be used to manage projects. The CPM and PM procedures are used to schedule tasks for a project subject to precedence, time, and resource constraints. The PM procedure is the interactive version of the CPM procedure. The Gantt procedure displays this schedule, and the Netdraw procedure displays the project network consisting of these tasks. These four procedures are integrated into the Projman application which is a friendly graphical user interface included as part of the SAS/OR software. This tutorial will cover the usage of these four procedures and the Projman application. The early part of the tutorial will cover the definition of terminology that is critical for understanding the output results. Then example projects containing resource, time, and precedence constraints will be scheduled using the Projman application. Finally two SAS macros that allow the conversion back and forth between data of Microsoft Project and the PM procedure will be discussed.

INTRODUCTION

The Project Management capabilities of SAS may be one of the best kept secrets of SAS/OR. It is useful for classroom instruction as well as tracking large administrative projects. It uses the CPM and PM procedures to schedule project activities that have precedence, resource, and time constraints. Using high or low resolution graphics, the system can construct Gantt Charts using the GANTT procedure and project networks using the NETDRAW. The PROJMAN menu system integrates these procedures and provides a syntax-free environment for project data management, scheduling, and report generation.

DEFINIITION OF PROJECT MANAGEMENT TERMS

The following terminology, as it relates to project management, is critical to understanding the output:

ACTUAL FINISH DATE

The calendar date work actually ended on an activity. It must be prior to the time now date.

ACTUAL START DATE

The calendar date work actually began on an activity. It must be prior to the time now date.

ALIGNMENT DATE

The calendar date an activity is desired to be started or completed; accepted as the date generated by the initial CPM schedule and resource allocation process.

CALENDAR

A calendar identifies project work days, and can be altered so weekends, holidays, vacation, weather days, etc., are not included.

CRITICAL PATH

The series of interdependent activities of a project, connected end-to-end, which determines the shortest total length of the project. The critical path of a project may change from time to time as activities are completed ahead of or behind schedule.

CRITICAL PATH METHOD (CPM)

The method used to determine the length of a project and to identify the activities that are critical to the completion of the project.

EARLY FINISH DATE (EF)

The earliest time an activity may be completed. It is equal to the early start time of the activity plus duration.

EARLY START DATE (ES)

The earliest time an activity may begin subject to any time constraints and the completion time of the preceding activity.

FREE FLOAT

The amount of time (in duration units) an activity may be delayed without affecting the early start of any of its immediate successor activities. It is equal to the difference between the early finish time of the activity and the early start time of the activity's immediate successors.

GANNT CHART

A graphic representation of work activities shown by a time-scaled bar chart.

LAG

The logical relationship between the start and/or finish of one activity and the start and/or finish of another activity.

The four basic types of lag relationships are:

1. Finish to Start
2. Start to Finish
3. Finish to Finish
4. Start to Start

LATE FINISH DATE (LF)

The latest time an activity may be completed without delaying the project finish date.

LATE START DATE (LS)

The latest time an activity may begin without delaying the project finish date. It is equal to the late finish time of the activity minus duration.

LOGIC

The interdependency of activities in the project network.

NETWORK DIAGRAM

A schematic display of the sequential and logical relationship of the activities which comprise the project.

RESOURCE CONSTRAINED SCHEDULING

The scheduling of activities in a project with the knowledge of certain resource constraints and requirements.

This process adjusts activity scheduled start and finish dates to conform to resource availability and use.

SCHEDULED FINIISH DATE (SF)

The date when the activity is scheduled to be completed using the resource constrained scheduling process.

SCHEDULED START DATE (SS)

The date when the activity is scheduled to begin using the resource constrained scheduling process. This date is equal to or greater than the early start date.

SUCCESSOR ACTIVITY

Any activity that exists on a common path with the activity in question and occurs after the activity in question.

TIME NOW DATE

The calendar date that separates actual (historical) data from scheduled data.

TOTAL FLOAT (TF)

The amount of time (in duration units) that an activity may be delayed from its early start time without delaying the project finish date. It is equal to the difference between the late finish time and early finish time (or the late start time and early start time) of the activity.

FEATURES OF THE PROJECT MANAGEMENT PROCEDURES

- a. Allows either activity-on-node or activity-on-arrow input.
- b. Generates schedules to meet project deadlines in a resource constrained environment.
- c. Generates calendar schedules.
- d. Generates Gantt Charts and Project Network diagrams in high resolution or low resolution graphics.
- e. Allows scheduling around holidays, weekends, and nonstandard days and weeks.
- f. Compares current progress of project with target schedule.
- g. Summarizes resource allocation and utilization.
- h. Generates resource requirements to meet critical deadlines.
- i. Allows activity splitting.
- j. Allows alternative resources.
- k. Generates PERT schedules.


EXAMPLE

Tom Lightner owns Lightner Construction, a general contracting company specializing in the construction of single-family residences and small office buildings. He wants to apply project management techniques to one of his home-building projects.

Activity	Description	Time Required (in days)	Immediate Predecessor Activities
A	Excavate	3	--
B	Lay foundation	4	A
C	Rough plumbing	3	B
D	Frame	10	B
E	Finish exterior	8	D
F	Install HVAC	4	D
G	Rough electric	6	D
H	Sheet rock	8	C, E, F, G
I	Install cabinets	5	H
J	Paint	5	H
K	Final plumbing	4	I
L	Final electric	2	J
M	Install flooring	4	K, L

The CPM procedure can be used to schedule the tasks(activities) that make up the entire project. These tasks have time and precedence constraints and may compete for scarce resources. It allows the scheduling of tasks around holidays and vacations. Both types of input: Activity-on-Node (AON) and Activity-on-Arc (AOA) are allowed.

Network Diagram –Activity on Node (AON)



In this diagram type, the activities are represented by the nodes, and the predecessor/successor relationships between activities are represented by the arrows. The following SAS program invokes the CPM procedure using AON representation.

```

/* Activity-on-Node representation of the project */
OPTIONS=110 PS=65 PAGENO=1;
DATA project;
  FORMAT task $16. succ1-succ3 $16.;
  INFILE cards MISSOEVER;
  INPUT @4 task & @20 days 2. @22 succ1 & @38 succ2 & @54 succ3 &;
  CARDS;
  Excavate          3 Lay foundation
  Lay foundation    4 Rough plumbing Frame
  Rough plumbing     3 Sheet rock
  Frame             10 Finish exterior Install HVAC Rough electric
  Finish exterior   8 Sheet rock
  Install HVAC      4 Sheet rock
  Rough electric    6 Sheet rock
  Sheet rock        8 Install cabinets Paint
  Install cabinets  5 Final plumbing
  Paint             5 Final electric
  Final plumbing     4 Install flooring
  Final electric    2 Install flooring
  Install flooring  4
;
RUN;
DATA holidays;
  holiday='02SEP13'd;
RUN;
ODS RTF; ODS graphics ON;
PROC CPM DATA=project DATE='1aug13'd INTERVAL=weekday HOLIDATA=holidays
OUT=results_AON;
  ACTIVITY task;
  DURATION days;
  SUCCESSOR succ1 succ2 succ3;
  HOLIDAY holiday;
  TITLE 'Lightner Construction';
  TITLE2 'Activity-on-Node Representation';
RUN;
PROC PRINT;
RUN;
ODS GRAPHICS OFF; ODS RTF CLOSE;
  
```


The following is the output in RTF format. The activities having zero float times are those that lie on the critical path. They have to be completed as scheduled in order for the project to be completed by October 4.

Obs	Task	succ1	succ2	succ3	days	E_START	E_FINISH	L_START	L_FINISH	T_FLOAT	F_FLOAT
1	Excavate	Lay foundation			3	01AUG13	05AUG13	01AUG13	05AUG13	0	0
2	Lay foundation	Rough plumbing	Frame		4	06AUG13	09AUG13	06AUG13	09AUG13	0	0
3	Rough plumbing	Sheet rock			3	12AUG13	14AUG13	03SEP13	05SEP13	15	15
4	Frame	Finish exterior	Install HVAC	Rough electric	10	12AUG13	23AUG13	12AUG13	23AUG13	0	0
5	Finish exterior	Sheet rock			8	26AUG13	05SEP13	26AUG13	05SEP13	0	0
6	Install HVAC	Sheet rock			4	26AUG13	29AUG13	30AUG13	05SEP13	4	4
7	Rough electric	Sheet rock			6	26AUG13	03SEP13	28AUG13	05SEP13	2	2
8	Sheet rock	Install cabinets	Paint		8	06SEP13	17SEP13	06SEP13	17SEP13	0	0
9	Install cabinets	Final plumbing			5	18SEP13	24SEP13	18SEP13	24SEP13	0	0
10	Paint	Final electric			5	18SEP13	24SEP13	20SEP13	26SEP13	2	0
11	Final plumbing	Install flooring			4	25SEP13	30SEP13	25SEP13	30SEP13	0	0
12	Final electric	Install flooring			2	25SEP13	26SEP13	27SEP13	30SEP13	2	2
13	Install flooring				4	01OCT13	04OCT13	01OCT13	04OCT13	0	0

The PROC CPM statement contains the start date of the project, August 1, 2013; the weekday (which excludes weekend days) as the time unit; and a reference to the holidays data set which includes Labor Day. Since the project will not be completed until after Labor Day, the tasks/activities are scheduled around this holiday. The Activity statement identifies the activity variable, Task. The Duration statement designates days as the duration variable. This variable measures the number of weekdays required to complete the associated task.

For AON representation, the immediate successors of each activity must be identified. These successor variables are contained in the Successor statement. Since none of the activities have more than three successors, we arbitrarily assigned them successor1, successor2, and successor3. Finally the Holiday statement identifies the variable in the dataset designated by the holiday option containing the date value for the Labor Day holiday.

Network Diagram –Activity on Arc (AOA)



This representation is not as popular as AON due to the necessary inclusion of dummy activities (represented by the dotted lines) having zero duration. The nodes represent benchmarks; in particular, Node 1 represents the beginning of the project, Node 13 the end. The predecessor/successor relationships among all the activities are given by their head node and tail node. The SAS program containing the CPM procedure using this network representation is given on the next page.

```

OPTIONS LS=110 PS=65 PAGENO=1;
DATA project;
  FORMAT task $16.;
  INFILE CARDS MISSOEVER;
  INPUT @4 task & @20 days tail head;
  CARDS;
  Excavate      3 1 2
  Lay foundation 4 2 3
  Rough plumbing 3 3 8
  Frame         10 3 4
  Finish exterior 8 4 5
  Install HVAC  4 4 6
  Rough electric 6 4 7
  Sheet rock    8 8 9
  Install cabinets 5 9 10
  Paint        5 9 11
  Final plumbing 4 10 12
  Final electric 2 11 12
  Install flooring 4 12 13
  Dummy1       0 5 8
  Dummy2       0 6 8
  Dummy3       0 7 8
;
DATA holidays;
  holiday='02SEP13'd;
run;
ODS RTF;
ODS graphics ON;
PROC CPM DATA=project DATE='1aug13'd INTERVAL=weekday HOLIDATA=holidays
OUT=results AOA;
  ACTIVITY task;
  DURATION days;
  TAILNODE tail;
  HEADNODE head;
  HOLIDAY holiday;
  TITLE 'Lightner Construction';
  TITLE2 'Activity-on-Arc Representation';
PROC PRINT;
RUN;
ODS GRAPHICS OFF;
ODS RTF CLOSE;

```

The TAILNODE and HEADNODE statement replace the SUCCESSOR statement used in the AON representation. The RTF output is on the following page.

Ob s	task	ta il	hea d	da ys	E_ STAR T	E_ FINI SH	L_ STAR T	L_ FINI SH	T_ FLOA T	F_ FLOA T
1	Excavate	1	2	3	01AUG13	05AUG13	01AUG13	05AUG13	0	0
2	Lay foundation	2	3	4	06AUG13	09AUG13	06AUG13	09AUG13	0	0
3	Rough plumbing	3	8	3	12AUG13	14AUG13	03SEP13	05SEP13	15	15
4	Frame	3	4	10	12AUG13	23AUG13	12AUG13	23AUG13	0	0
5	Finish exterior	4	5	8	26AUG13	05SEP13	26AUG13	05SEP13	0	0
6	Install HVAC	4	6	4	26AUG13	29AUG13	30AUG13	05SEP13	4	0
7	Rough electric	4	7	6	26AUG13	03SEP13	28AUG13	05SEP13	2	0
8	Sheet rock	8	9	8	06SEP13	17SEP13	06SEP13	17SEP13	0	0
9	Install cabinets	9	10	5	18SEP13	24SEP13	18SEP13	24SEP13	0	0
10	Paint	9	11	5	18SEP13	24SEP13	20SEP13	26SEP13	2	0
11	Final plumbing	10	12	4	25SEP13	30SEP13	25SEP13	30SEP13	0	0
12	Final electric	11	12	2	25SEP13	26SEP13	27SEP13	30SEP13	2	2
13	Install flooring	12	13	4	01OCT13	04OCT13	01OCT13	04OCT13	0	0
14	Dummy1	5	8	0	06SEP13	06SEP13	06SEP13	06SEP13	0	0
15	Dummy2	6	8	0	30AUG13	30AUG13	06SEP13	06SEP13	4	4
16	Dummy3	7	8	0	04SEP13	04SEP13	06SEP13	06SEP13	2	2

GANTT Procedure


After the CPM procedure has created the project schedule, the GANTT procedure can be used to display a Gantt chart. This is a graphical scheduling tool for the planning and control of a project. A Gantt chart based on the AON network representation can be easily generated from the following code:

```
PROC GANTT GRAPHICS DATA=results_AON;
ID task;
TITLE 'Lightner Construction';
TITLE2 'Activity-on-Node Representation';
RUN;
```

where `results_AON` is the schedule data set produced from the CPM procedure.

The Gantt Chart produced is on the next page:

Lightner Construction Activity-on-Node Representation




The activities highlighted entirely in green are the critical activities that form the critical path.

Likewise a Gantt chart based on the AOA representation could be produced using similar code:

```
PROC GANTT GRAPHICS DATA=results_AOA;
ID task;
TITLE 'Lightner Construction';
TITLE2 'Activity-on-Arc Representation';
RUN;
```

The Gantt chart appears on the following page:

Lightner Construction Activity-on-Arc Representation



LEGEND: ■ Duration of a Normal Job ■ Slack Time for a Normal Job ■ Duration of a Critical Job


NETDRAW Procedure

The NETDRAW procedure can be used to draw a network diagram of the activities making up the project. It is a general tool that can be used to draw any network, so it does not necessarily require the output from the CPM procedure. The following program will produce an AON representation of the project using high resolution graphics:

```

GOPTIONS HPOS=100 VPOS=70 BORDER FS;
PATTERN1 C=green V=e;
TITLE H=3 J=1 F=swiss 'Lightner Construction-Activity on Node Representation';
PROC NETDRAW GRAPHICS DATA=results_AON;
ACTNET / ACT=task
      SUCC=(succ1 succ2 succ3)
      CARCS=blue
      LWIDTH=3
      PCOMPRESS
      DUR=days;
RUN;
```

Lightner Construction—Activity on Node Representation



The GOPTIONS statement sets the global options HPOS= and VPOS= to 100 and 70 respectively. These options control the number of character cell positions on the screen. The PATTERN statement specifies that the color of the nodes is green. The color of the arcs as specified by the CARCS= option is blue. The LWIDTH= option sets the width of all lines as 3; the FONT= option specifies the font for the text to be SWISS; and the PCOMPRESS option compresses the entire diagram to one page.

PM Procedure

Unlike the CPM procedure, the PM procedure is interactive. It has many of the same scheduling features of CPM but only supports AON input. One can interactively add activities, change the duration of activities, and change precedence constraints between activities. To invoke this procedure, the project data set created for the CPM procedure is the input data set. For example the following code would launch this procedure.

```

PROC PM DATA=project OUT=results2
  INTERVAL=weekday
  HOLIDATA=holidays
  DATE='01aug13'd;
  ACTIVITY task;
  DURATION days;
  HOLIDAY holiday;
  SUCCESSOR succ1 succ2 succ3;
  RUN;
  
```

The syntax for PM is similar to that of CPM. An interactive window will launch with a Gantt chart displaying all the activities. The chart automatically updates when any change is made to an activity.

PROJMAN MENU SYSTEM

The Project Management (PROJMAN) Menu System is a graphical interface for manipulating project data, creating project schedules, and generating reports. It integrates the CPM, GANNT, NETDRAW, and PM procedures and provides a syntax-free environment for project data management, scheduling, and report generation. The system provides a collection of features, including BY and WHERE processing, the ability to view and save results, extensive facilities for report customization, an online tutorial, and generous context-specific help.

To access this system, from the main menu, select
Solutions -> Analysis-> Project Management

MICROSOFT PROJECT CONVERSION MACROS

There are two SAS macros that convert Microsoft Project data back and forth to SAS data. The macro %MSPTOSAS converts Microsoft Project 98, 2000, 2002, 2003, and 2007 data to a form readable by the PM procedure. Execution of this macro requires SAS/ACCESS® Interface to PC File Formats. The %SASTOMP macro converts data sets used by the CPM and PM procedures into a form readable by Microsoft Project 2000, 2002, and 2003.

The following are examples that use these two macros. We begin with the following SAS code that uses data from the Lightner Construction example:

* This file reference contains the directory where the Microsoft project file will reside;

```
FILENAME msproj 'F:\Project Management\sasmsp.mdb';
DATA activity;
  FORMAT activity $16. succ1-succ3 $16.;
  INFILE CARDS MISSOVER;
  INPUT @4 activity & @20 dur 2. @22 succ1 & @38 succ2 & @54 succ3 &;
  CARDS;
  Excavate          3 Lay foundation
  Lay foundation   4 Rough plumbing Frame
  Rough plumbing   3 Sheet rock
  Frame           10Finish exterior Install HVAC Rough electric
  Finish exterior 8 Sheet rock
  Install HVAC     4 Sheet rock
  Rough electric  6 Sheet rock
  Sheet rock      8 Install cabinets Paint
  Install cabinets5 Final plumbing
  Paint           5 Final electric
  Final plumbing  4 Install flooring
  Final electric  2 Install flooring
  Install flooring4
;
RUN;
DATA holidays;
  HOLIDAY='02SEP13'd;
RUN;
PROC PM DATA=activity;
  ACT activity;
  SUCC succ1 succ2 succ3;
  DURATION dur;
RUN;
```

The Macro statements are on the following page.

```
/* Macro call will create the Microsoft Project 2003 file */
%sastomsp (mdbfile=msproj)
/* Macro call will convert the Microsoft Project 2003 file back to a file readable by PM procedure */
%MSPTOSAS ( LIBRARY='F:\Project Management', MDBFILE=msproj, VERSION=2003)
```

There are more extensive examples in the online SAS Help and Documentation.

SAMPLE PROJECT MANAGEMENT PROGRAMS

There are several SAS sample programs pertaining to Project Management available for your use. To access these programs:

1. Select **Help** from the SAS menu.
2. Select **SAS Help and Documentation**
3. Select **Learning to Use SAS → Sample SAS Programs**
4. Select **SAS/OR → Samples**
5. The next screen should list a Table of Contents of SAS/OR programs.

CONCLUSIONS

This tutorial provided an introduction to the project management terminology and procedures as well as the Project Management features of SAS/OR. SAS does provide a powerful command set that not only allows project control but also high resolution graphics that display an exciting view of the project from start to completion. SAS Macros have been developed to convert a SAS project into Microsoft project format and a Microsoft project back into SAS format.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.

REFERENCES

1. Spreadsheet Modeling & Decision Analysis, Cliff T. Ragsdale, 4th edition, Thomson Southwestern.

CONTACT INFORMATION:

Dr. Jon Patton
Senior Teaching Technologies Analyst
Miami Computer and Information Service
3071 Laws Hall
E-mail: pattonjm@miamioh.edu
Office Phone:513-529-8751
Fax:513-529-5577